

1 Uso de SPICE (operación)

1.1 Introducción

El programa PSpice, y en general cualquier programa SPICE, recibe la información del circuito que tiene que simular y del tipo de análisis que tiene que realizar a través de un fichero fuente o fichero de circuito. Este fichero contiene una descripción del circuito, de los componentes que lo forman y de la interconexión entre ellos llamada *netlist*. Además de la *netlist* en el fichero de circuito se indican las clases de análisis que se han de efectuar y las directrices de dichos análisis. La *netlist* sigue una cierta sintaxis, adecuada para ser interpretada correctamente por el programa de análisis SPICE (en el apéndice H se detallan las convenciones que ha de seguir el fichero de circuito).

La generación de este fichero de circuito o fichero fuente –ambas denominaciones se utilizarán indistintamente– se puede hacer de diversas formas. Puede haberse generado como resultado de un programa de captura de esquemáticos, que es un programa gráfico en el que se introduce un circuito y que es capaz de generar un fichero con el *netlist* correspondiente al circuito dibujado, puede provenir de un programa de diseño a nivel de componentes, etc.. Pero la forma más sencilla, que será la que se tratará en este manual, es la generación directa mediante un editor de textos adecuado.¹ Aunque puede tener cualquier nombre es conveniente que el fichero de circuito tenga la extensión *.CIR* ya que ésta es la extensión que reconoce por defecto el programa PSpice.

Una vez generado el fichero que contiene la *netlist* se inicia el análisis ejecutando el programa PSPICE1. Se puede llamar a este programa sin parámetros; en este caso se solicitará el nombre del fichero fuente así como el nombre del fichero de salida. Otra opción es pasar como parámetro el nombre del fichero fuente y, si se desea, el nombre del fichero de salida. Tanto en un caso como en el otro se realizará el análisis y se guardarán los resultados en el fichero de salida. Si no se había asignado ningún nombre al fichero de salida se tomará por defecto el nombre del fichero fuente con la extensión *.OUT*.

¹ El editor utilizado ha de ser del tipo programación, es decir, que no introduzca ningún carácter o control extra. Son válidos EDIT, EDLIN del MS-DOS, etc.

El fichero de salida *.OUT* contiene información de cómo se ha realizado el análisis y los resultados obtenidos. Incluye el *netlist*, el tiempo de ejecución, y, según el caso, información del punto de reposo, etc. Si el análisis no se ha podido realizar con éxito, en este fichero se indicará la causa. Es un fichero tipo texto y se puede visualizar simplemente mediante la orden *TYPE* o usando cualquier editor de textos.

1.2 Estructura del fichero del circuito

Como se ha mencionado anteriormente el fichero circuito o fuente contiene el *netlist* junto con una serie de directrices orientadas a indicar tipo y entorno del análisis. Por ello en el fichero de circuito se encontrarán tres tipos de líneas: *a)* las que describen el circuito y que conforman la *netlist*; *b)* las líneas de comandos: comienzan con un punto y se utilizan para definir los tipos de análisis que se deben efectuar y como sentencias de control ; *c)* las líneas de comentarios.

a) Líneas correspondientes al *netlist*:

Definen los elementos que componen el circuito tales como resistencias, condensadores, transistores, etc., así como las fuentes de tensión, de corriente, etc. El tipo de dispositivo de que se trata queda determinado por la primera letra. Así, si el nombre del dispositivo empieza por R definirá una resistencia, si por una C se tratará de un condensador, etc.. El nombre siempre debe comenzar por una letra permitida (que defina uno de los componentes existentes) y continuar con una secuencia de letras, números o los siguientes caracteres: "_", "/", "*", "%", "\$". La extensión máxima es de 131 caracteres, pero se recomienda que no sobrepasen los 8 caracteres. Algunos de los elementos disponibles en SPICE son:

R ≡ Resistencias

C ≡ Condensadores

L ≡ Inductancias

V ≡ Fuentes independientes de tensión

I ≡ Fuentes independientes de corriente

En los capítulos 3 y 4 se tratan con más detalle los dispositivos más comunes y en el apéndice C se recoge un resumen.

Tras el nombre se ha de indicar los nodos entre los que está conectado el elemento, que como mínimo han de ser dos, y por último el valor del componente.² Para ello, antes de introducir el circuito es

² Esta es la estructura más sencilla, pero las líneas pueden ser más complicadas.

necesario nombrar los nodos del circuito. Por ejemplo, supóngase que el circuito que se desea analizar es el divisor de tensión de la figura 1.1.




Fig. 1.1

El primer paso consiste en etiquetar los nodos. Los nodos se etiquetan mediante números enteros positivos. No es necesario que sean correlativos, pero es obligatorio etiquetar un nodo con el número cero. Este nodo será el que se tomará como referencia o masa.³

A continuación se puede introducir el circuito. El circuito propuesto se podría definir mediante las siguientes líneas:

```
R1 1 0 4000
```

Con esta línea se indica que hay una resistencia, a la que se le asignará el nombre de R1, conectada entre los nodos 1 y cero (masa) con un valor de 4000 ohmios. Esto es así porque el nombre del componente comienza con R e indica que se trata de una resistencia. A continuación se han entrado los nodos entre los que está conectada, separados por un espacio. Finalmente se ha especificado el valor del componente. Las unidades dependen del componente y son siempre las fundamentales, en este caso serán ohmios.

La siguiente resistencia se describe:

```
RSUPERIOR 2 1 2K
```

En este caso se ha preferido dar un nombre más significativo al componente, aunque cualquier nombre es válido siempre que empiece con R ya que es una resistencia. Seguidamente se indica que está conectada entre los nodos 2 y 1 y que posee un valor de 2000 ohmios.

³ PSpice, la versión de Microsim de SPICE, admite que se etiqueten los nodos, excepto el nodo cero, con cualquier cadena de caracteres.

Como se puede observar se admiten prefijos para designar el valor de los componentes. Los prefijos aceptados se indican en el apéndice G. Deben estar situados inmediatamente después del valor del componente, sin ningún espacio de separación.

La fuente de tensión independiente se define y conecta de la siguiente forma:

VBAT 2 0 6

La V inicial de la palabra VBAT indica que se trata de una fuente de tensión independiente. Está conectada entre los nodos 2 y cero y posee un valor de 6 voltios.

Las expresiones que se han empleado corresponden a la formulación más sencilla. Tal como se verá a lo largo de la presente publicación, las sintaxis completas son más complicadas. En el caso de una fuente de tensión la sintaxis completa, siguiendo las convenciones explicadas en el prólogo, es:

V<nombre> <nodo+> <nodo-> [[DC] <valor>] [AC <magnitud> [fase]] [transitorio]

Cada uno de los corchetes se aplica a uno de los tipos de análisis que se estudiarán posteriormente, pero si no se especifica nada se considera que el valor proporcionado es un valor para el análisis ".DC". Por ello la anterior definición de VBAT es equivalente a:

VBAT 2 0 DC 6

Con objeto de hacer más inteligible el circuito se puede añadir una letra para indicar las unidades, pero ésta ha de venir inmediatamente después del valor (o del prefijo en caso de que se utilice) por lo que no puede haber ningún espacio entre el valor y la unidad. De haber usado esta posibilidad la línea anterior quedaría:

VBAT 2 0 DC 6V

El programa lee los números que indican el valor del componente, si el carácter que viene a continuación corresponde a un prefijo lo interpreta como tal y todos los demás números o caracteres hasta el próximo espacio en blanco son ignorados.

Se ha de prestar especial cuidado en el uso de los faradios ya que la letra F se confundirá con el prefijo *femto*; por ello no se puede especificar las unidades en las capacidades. Por ejemplo, la línea:

CEX 4 5 20 crea una capacidad entre los nodos 4 y 5 de 20 faradios.

y

CEX 4 5 20F se refiere a una capacidad de 20 femto-faradios.

Un aspecto importante es el convenio de signos, es decir, la polaridad. La polaridad queda determinada mediante el siguiente convenio: el primer nodo (en el caso de VBAT el nodo 2) se con-

sidera el nodo positivo y el segundo el negativo. El sentido positivo de corriente entra por el nodo positivo. Según este convenio la corriente a través de la fuente de tensión del circuito que se está analizando será negativa. Este convenio se aplica a cualquier elemento de dos terminales, lo que puede dar lugar a confusiones especialmente con las fuentes de corriente.

b) Tipos de análisis:

Son líneas de comandos que indican los análisis que se desean realizar sobre el circuito. Como todas las líneas de comando comienzan con un punto. Se pueden hacer análisis de respuesta en frecuencia en régimen permanente, análisis de respuesta transitoria, etc. Siempre se hace un análisis del punto de reposo, aun en el caso que no se solicite ningún análisis.

Los diversos análisis que se pueden realizar sobre el circuito ya introducido se estudiarán en detalle en el capítulo 2, pero para poder realizar un ejemplo inicial completo se examinará la siguiente sentencia que originará un análisis de barrido en continua (*DC Sweep*):

```
.DC VBAT 1 5 0.5
```

Como esta línea es un comando empieza por un punto. Con ella se indica que se desea analizar el circuito para diversos valores de la fuente de tensión VBAT. El primer número es el valor inicial, el segundo el valor final y el tercero el paso. Así pues, como consecuencia de esta sentencia se analizará el circuito para un valor de VBAT de 1V, seguidamente se hará un segundo análisis con VBAT de 1,5V y así sucesivamente hasta que VBAT alcance el valor de 5V.

Sobre un mismo circuito se pueden realizar más de un tipo de los análisis que se estudiarán posteriormente.

c) Sentencias de control:

Por ser líneas de comandos comienzan con un punto. Realizan diversas funciones como definir modelos, especificar condiciones iniciales, guardar resultados, definir el entorno, etc. Algunas de ellas irán apareciendo paulatinamente y otras se recogen en los apéndices. Por ahora sólo se utilizará la sentencia ".END" utilizada para señalar el final de un circuito ya que, puesto que en un fichero de circuito se permite definir más de un circuito, para separarlos se utiliza la sentencia ".END".

d) Comentarios:

Con objeto de clarificar el circuito se puede, y es altamente recomendable, hacer uso de las líneas de comentarios. Las líneas de comentarios se indican mediante un asterisco "*" en la primera columna. También es posible introducir comentarios en mitad de una línea normal mediante un ";". Todo lo que venga a continuación del punto y coma será ignorado por el programa.

La primera línea del fichero siempre se considera como línea de comentario aunque no empiece con asterisco y además se utiliza como cabecera en la presentación de los datos. Se ha de tener cuidado

con esta característica de la primera línea, pues se puede cometer el error de utilizar dicha línea para introducir algún elemento del circuito o alguna sentencia de control que serían ignorados.

1.3 Ejecución del análisis y resultados

Una vez que ya se dispone del fichero de circuito se puede realizar su análisis y examinar los resultados. Si no se ha especificado ningún análisis el programa se limita a calcular el punto de reposo, es decir, calcula las tensiones en cada nodo del circuito y las corrientes en las ramas, utilizando los valores especificados de las fuentes independientes. El resultado se guarda en el fichero de salida que normalmente tiene el mismo nombre y está situado en el mismo directorio que el fichero fuente, pero posee la extensión ".OUT". Se pueden obtener resultados más completos mediante las órdenes ".PRINT" y ".PLOT" o bien ejecutando un programa auxiliar de representación llamado PROBE.

El siguiente ejemplo mostrará el proceso que debe seguirse de inicio a fin. Si se desea analizar el divisor de tensión propuesto en la figura 1.1 el circuito fuente sería:

EJERCICIO 1.1

```
* Elementos del circuito
R1 1 0 4000 ; resistencia de 4K
RSUPERIOR 2 1 2K ; resistencia de 2K
VBAT 2 0 6 ; fuente de tensión

.END
```

La primera línea *-EJERCICIO 1.1-* es un comentario. Al ser la primera línea no es preciso iniciarla con un asterisco y además se tomará como cabecera en la presentación de resultados. A continuación viene la descripción del circuito y finalmente se señala el final del mismo con el comando *.END*.

Para efectuar el análisis se ha de ejecutar el siguiente programa desde la línea de comandos del DOS:

```
PSPICE14
```

A continuación el programa solicitará el fichero fuente y el fichero de salida (éste último es opcional). También se puede especificar el fichero fuente como primer parámetro y el fichero de salida como segundo parámetro, por ejemplo:

```
PSPICE1 EX1_1 SALIDA
```

⁴ Estos pasos se pueden realizar también haciendo uso del entorno integrado que se verá al final de este capítulo.

donde se supone que el fichero fuente se llama EX1_1.CIR y se desea que el de salida se llame SALIDA.OUT. Nótese que las extensiones vienen dadas por defecto y recuérdese que si no se proporciona el nombre del fichero de salida éste tomará el nombre del fichero fuente.

Mientras se ejecuta en la pantalla se indica el análisis que se está realizando. En este caso como no se ha solicitado ningún tipo de análisis, se calcula exclusivamente el punto de reposo. Al acabar se puede consultar el resultado en el fichero de salida. Su contenido será el siguiente:

```

**** 07/28/94 20:05:38 ***** Evaluation PSpice (September 1991)
EJERCICIO 1.1

****   CIRCUIT DESCRIPTION
*****
* Elementos
R1 1 0 4000 ; resistencia de 4K
RSUPERIOR 2 1 2K ; resistencia de 2K
VBAT 2 0 6 ; fuente de tensión

.END

**** 07/28/94 20:05:38 ***** Evaluation PSpice (September 1991)
EJERCICIO 1.1

****   SMALL SIGNAL BIAS SOLUTION   TEMPERATURE = 27.000 DEG C
*****
NODE   VOLTAGE NODE   VOLTAGE NODE   VOLTAGE
( 1)   4.0000  ( 2)   6.0000

VOLTAGE SOURCE CURRENTS
NAME    CURRENT

VBAT    -1.000E-03

TOTAL POWER DISSIPATION 6.00E-03 WATTS

JOB CONCLUDED

TOTAL JOB TIME      1.04

```

Se observa cómo tras reproducir el fichero de entrada se muestra el valor del punto de reposo - *SMALL SIGNAL BIAS SOLUTION*- con la tensión en cada nodo, la corriente a través de las fuentes independientes de tensión y, por último, la potencia consumida.

Lo habitual será especificar algún tipo de análisis. En este caso inicialmente se realiza el análisis del punto de reposo como en el caso anterior y a continuación el análisis o análisis especificados. Pero para guardar los resultados de los análisis es necesario indicar qué variables se desean mediante uno de los siguientes comandos: ".PRINT", ".PLOT" o ".PROBE".

El comando *.PRINT* proporciona una tabla de valores en el fichero de salida. La sintaxis básica es la siguiente:

.PRINT <tipo de análisis> <lista de variables>

En esta sentencia se indica, mediante <lista de variables>, cuáles son las variables con las que se construirá la tabla. Como ya se ha comentado es posible especificar más de un análisis sobre un mismo circuito. Por ello se ha de especificar sobre qué tipo de análisis se desean grabar los resultados. Si se desean guardar los resultados de más de un análisis será necesario incluir una sentencia ".PRINT" por cada análisis. El comando sólo afecta a la obtención de los resultados y no al análisis en sí mismo. De esta forma, aunque se solicite sólo el valor de una variable se calculan todas las variables.

A modo de ejemplo supóngase que se desea un estudio del comportamiento del circuito anterior para diversos valores de la tensión VBAT y que, en concreto, interesa conocer el valor que toma la tensión en el nodo 1. Para conseguirlo será necesario realizar un análisis tipo ".DC" e incluir una sentencia ".PRINT" que indique las variables que se desean consultar. El fichero fuente podría ser:

```

EJERCICIO 1.2

* Elementos
R1 1 0 4000 ; resistencia de 4K
RSUPERIOR 2 1 2K ; resistencia de 2K
VBAT 2 0 6 ; fuente de tensión

.DC VBAT 1 5 .5 ;análisis variando el valor de VBAT
                ;entre 1 y 5 voltios con incremento de 0,5V

.PRINT DC V(1) I(R1) ;se hace una tabla con la tensión entre el
                    ;nodo 1 y la masa y la corriente a través de R1
                    ;obtenidas en el análisis DC

.END

```

La sentencia ".DC" indica qué análisis se desea realizar y ya se explicó en la página 15. La sentencia ".PRINT" establece que se ha de realizar una tabla con los valores que toma la tensión en el nodo 1 y la corriente a través de la resistencia R1 durante el análisis en DC.

Las variables pueden ser tensiones entre dos nodos o corrientes a través de un dispositivo de dos terminales. El tipo de variables de salida permitidos y la forma de especificarlas están recogidas en el apéndice F. Para especificar una tensión entre dos nodos se ha de introducir: $V(\text{nodo \#}, \text{nodo \#})$. Es posible indicar un único nodo, en este caso el valor que se obtendrá será la tensión entre dicho nodo y la masa (nodo cero). Otra posibilidad es poner dentro del paréntesis el nombre de un dispositivo de dos terminales, con lo que se imprimirá la tensión sobre dicho dispositivo. Por ejemplo, en el circuito introducido:

V(1) es equivalente a V(1,0) y a V(R1)

y

V(2) es equivalente a V(2,0) y a V(RSUPERIOR)

Una corriente se especifica mediante: $I(\text{dispositivo})$, donde *dispositivo* ha de ser un elemento de dos terminales. Según esta norma en el ejemplo se está solicitando la tensión en el nodo 1 con respecto a la masa y la corriente sobre la resistencia R1 durante un análisis tipo DC en que se dan valores a VBAT entre 1 y 5 voltios. El resultado puede verse en el fichero de salida ".OUT", que en nuestro caso sería:

```

**** 07/29/94 15:44:03 ***** Evaluation PSpice (September 1991) *****

EJERCICIO 1.1
****   CIRCUIT DESCRIPTION
*****
* Elementos

R1 1 0 4000 ; resistencia de 4K
RSUPERIOR 2 1 2K ; resistencia de 2K
VBAT 2 0 6 ; fuente de tensión

.DC VBAT 1 5 .5 ;un análisis variando el valor de VBAT
                ;entre 1 y 5 voltios con incremento de 0,5V

.PRINT DC V(1) I(R1) ;se hace una tabla con la tensión entre el
                    ;nodo 1 y la masa y la corriente a través de R1
                    ;obtenidas en el análisis DC

.END

**** 07/29/94 15:44:03 ***** Evaluation PSpice (September 1991)
EJERCICIO 1.1
****   DC TRANSFER CURVES           TEMPERATURE = 27.000 DEG C
*****
      VBAT           V(1)           I(R1)

```

1.000E+00	6.667E-01	1.667E-04
1.500E+00	1.000E+00	2.500E-04
2.000E+00	1.333E+00	3.333E-04
2.500E+00	1.667E+00	4.167E-04
3.000E+00	2.000E+00	5.000E-04
3.500E+00	2.333E+00	5.833E-04
4.000E+00	2.667E+00	6.667E-04
4.500E+00	3.000E+00	7.500E-04
5.000E+00	3.333E+00	8.333E-04
JOB CONCLUDED		
TOTAL JOB TIME		1.87

NOTA: Es de destacar que en la tabla no aparece el valor de V(1) correspondiente a VBAT igual a 6 voltios. Esto es debido a que al hacer el análisis DC y utilizar los valores especificados en dicho análisis es indiferente el valor que se haya dado a la fuente de tensión VBAT. El valor de VBAT se utiliza sólo para el cálculo del punto de reposo.

Si además de los resultados de los análisis se precisa información detallada del punto de reposo se puede incluir la sentencia ".OP".

El comando ".PLOT" es similar a ".PRINT" pero proporciona una representación gráfica en lugar de una tabla. La representación se hace mediante caracteres, lo que es adecuado para cualquier tipo de impresora aunque ésta no tenga capacidades gráficas pero, por contra, los resultados son muy pobres. El método más eficaz y cómodo de obtener una representación gráfica es mediante el comando ".PROBE" con lo que queda reservado el uso de ".PRINT" y ".PLOT" a situaciones en las que se requiera una salida tabulada (caso de PRINT) o en las que no se disponga de una tarjeta gráfica adecuada (caso de PLOT).

Mediante el comando ".PROBE" se guardan los resultados de los análisis ".DC", ".AC" y transitorio (".TRAN") que se hayan especificado, en un fichero con extensión ".DAT". Por defecto el nombre de este fichero es el mismo que el del fichero fuente.

Los datos que se almacenan son las tensiones de *todos* los nodos y las corrientes a través de *todos* los dispositivos. Por ello, a diferencia de los comandos ".PRINT" y ".PLOT", no es necesario indicar las variables que se desea guardar. Pero si se prevé que el fichero generado será muy grande se pueden limitar las variables que se almacenarán indicándolas después del comando ".PROBE", aunque, en circuitos normales con análisis normales no suele ser necesario ya que el tamaño del fichero suele ser aceptable.

El siguiente fichero de circuito hace uso de PROBE:

EJERCICIO 1.1

* Elementos

R1 1 0 4000 ; resistencia de 4K
RSUPERIOR 2 1 2K ; resistencia de 2K
VBAT 2 0 6 ; fuente de tensión

.DC VBAT 1 5 .5 ; análisis variando el valor de VBAT
; entre 1 y 5 voltios con incremento de 0,5V

.PROBE

.END

El fichero generado como consecuencia de la instrucción ".PROBE" –extensión ".DAT"– es un fichero binario (aunque se puede especificar que sea de tipo texto: opción CSDF) lo que implica que no se puede visualizar mediante un comando TYPE o un editor de textos. Para visualizar los datos se tiene que ejecutar, a posteriori, un programa específico que sea capaz de interpretarlos. Este programa es el PROBE (no confundir con el recién visto comando ".PROBE"). Al acabar la ejecución del PSPICE1 si se desea visualizar los resultados de los análisis se ha de ejecutar el programa PROBE pasando como parámetro el nombre del fichero ".DAT" que se desea procesar.

Habitualmente el fichero que interesa es el correspondiente al último análisis realizado. En este caso se pueden realizar estos pasos automáticamente ya que se dispone a este fin de un fichero ".BAT", llamado PSPICE.BAT. Si se ejecuta este programa, primero se ejecutará el programa de análisis PSPICE1 y al acabar se llamará al programa PROBE, pasándole como parámetro el fichero ".DAT" generado durante el análisis.⁵

De forma somera la utilización de PROBE es la siguiente:

Una vez se ha entrado en el programa hay que elegir el tipo de análisis del que se quiere consultar los resultados. Si sólo se ha realizado un análisis se salta este paso. A continuación se selecciona la variable que se desea visualizar. Esto se consigue mediante la orden "Add_trace" (presionando F4 o el botón derecho del ratón aparece una lista con todas las variables disponibles). Las sintaxis para las variables es similar a la utilizada en el comando

⁵ Cuando se ejecuta PROBE éste busca el nombre del fichero ".DAT" que se quiere procesar en el fichero PROBE.NAM. Sólo si el fichero no existe pedirá el nombre del fichero que se quiere procesar. El proceso lanzado por PSPICE.BAT se encarga, entre otras cosas, de que PROBE.NAM contenga el nombre del último análisis realizado por lo que estos pasos se realizan automáticamente.

".PRINT" y se detallan en el apéndice F. Se puede especificar más de una variable en cada orden "Add_trace". Cada variable se representará mediante una línea llamada traza. Para obtener información numérica de un punto de una traza pueden utilizarse los cursores.




Fig. 1.2

Los pasos anteriores producirían la gráfica mostrada en la figura 1.2, en la que se han visualizado las tensiones VBAT y V(1).

Si se desea información más completa se puede consultar el capítulo 5 dedicado al programa PROBE.

1.4 Parámetros y topología

1.4.1 Parámetros

Es posible definir constantes que serán usadas posteriormente en diversos puntos del circuito para dar el valor de los componentes. En SPICE estas constantes se llaman parámetros globales⁶ y se declaran mediante la sentencia ".PARAM" según la siguiente sintaxis:

```
.PARAM <nombre> = <valor>
```

⁶ Se llaman globales para distinguirlos de los parámetros de los modelos de los que se hablará más adelante.

ejemplo:

```
.PARAM CONST = 2K
```

Ahora se puede utilizar el parámetro CONST como valor de un componente. Esta posibilidad es útil por diversos motivos. Si en un circuito existen varios componentes que poseen el mismo valor éste puede ser asignado usando el parámetro. Así la resistencia RSUPERIOR de un valor de 2 K Ω se podría haber definido de la forma siguiente:

```
RSUPERIOR 2 1 {CONST}
```

Cualquier otra resistencia del mismo valor se podría declarar de la misma forma. Si en un momento determinado se decide que las resistencias han de tener un valor diferente bastará con modificar una única línea, la línea de definición del parámetro.

Obsérvese la utilización de las llaves: éstas son obligatorias para que se interprete correctamente el valor del parámetro. De hecho las llaves designan una expresión o función matemática. La forma mostrada es la más sencilla en que la función es simplemente una constante, pero se admiten expresiones mucho más complejas. Las funciones disponibles se recogen en el apéndice E.⁷

El uso de expresiones muestra otra ventaja en el uso de parámetros ya que, mediante su uso, se simplifica la asignación y la posterior modificación de valores de componentes con valores relacionados entre sí. Así, en el ejemplo, la resistencia R1 se podría haber declarado:

```
R1 1 0 {2*CONST}
```

Pero la ventaja fundamental del uso de parámetros es que se puede ordenar un análisis ".DC" tomando como variable un parámetro. Supóngase que se desea analizar la respuesta del circuito anterior para diversos valores de resistencias. Si no se usan parámetros será necesario crear un fichero fuente para cada valor de la resistencia y realizar un análisis para cada circuito. Con el uso de los parámetros se podrá realizar un análisis ".DC" variando el parámetro. Al estudiar el análisis ".DC" se comentará este extremo.

1.4.2 Restricciones topológicas

Se han de cumplir una serie de normas en la descripción del circuito para que éste sea interpretado correctamente. Estas normas son:

⁷ Estas mismas funciones se pueden usar, en su mayoría, en el programa PROBE.

- a) Todos los nodos deben tener como mínimo dos ramas. SPICE comprueba el circuito antes de comenzar ningún análisis y si detecta un error de este tipo genera el siguiente mensaje en el fichero de salida:

ERROR -- Less than 2 connections at node 2

- b) No deben existir nodos flotantes. Se entiende por nodo flotante todo aquel que no tenga un camino *en continua* a la masa (nodo cero). Según esto el circuito de la figura 1.3 produciría un error que SPICE indicaría con el mensaje

ERROR -- Node is floating




Fig. 1.3

- c) Se han de evitar los bucles de tensión. Se produce un bucle de tensión cuando existe un lazo o una malla compuesta exclusivamente de fuentes de tensión. SPICE avisa de esta situación con:

ERROR -- Voltage source and/or inductor loop involving ...

You may break the loop by adding a series resistance

Si bien estas normas, a primera vista, son fáciles de seguir, se pueden violar inadvertidamente al utilizar determinados dispositivos. Por ejemplo, el circuito de la figura 1.4 es incorrecto ya que el nodo 2 es un nodo flotante. Efectivamente, en continua los condensadores se comportan como circuitos abiertos por lo que no existe un camino de continua desde el nodo 2 al nodo cero.

Este problema se puede resolver añadiendo una resistencia de valor muy alto, en paralelo con el condensador. En el circuito de la figura 1.5 se ha utilizado esta solución y de esta manera se proporciona un camino a la masa. La resistencia ha de ser de un valor tal que no afecte al comportamiento del circuito (resistencias del orden de $10^9\Omega$ pueden ser adecuadas).




Fig. 1.4




Fig. 1.5

La otra regla –prevención de bucles de tensión– puede violarse inadvertidamente al usar inductancias. El programa modela las inductancias como fuentes de tensión controladas. Esto significa que dos inductancias pueden provocar bucles de tensión allá donde, en principio, no parecía existir tal problema.

En el circuito de la figura 1.6 se presenta un caso de bucle de tensión provocado por la existencia de dos inductancias (dos fuentes de tensión) conectadas en paralelo. El mismo error se produciría si se conectase una inductancia con otro tipo de fuente de tensión.




Fig. 1.6

En estas circunstancias se puede conectar una resistencia en serie con la inductancia, y romper así el mencionado bucle. La resistencia añadida sustituye a un cortocircuito, por tanto, para que no tenga influencia sobre el comportamiento del circuito, ha de ser de un valor muy bajo. Valores del orden $0,001\Omega$ pueden ser adecuados. Esta solución se ha aplicado en el circuito de la figura 1.7.




Fig. 1.7

1.5 Entorno de simulación

Hay un tipo de órdenes que están destinadas a definir el modo en que se realizará la simulación (precisión de los cálculos, temperatura, valores por defecto de algunos parámetros, etc.), la cantidad de información que se proporcionará, la situación de las librerías de los modelos y los subcircuitos, etc. Algunas de estas órdenes son:

.OP : proporciona información completa sobre el punto de reposo. Se detalla la potencia disipada y las corrientes sobre todas las fuentes de tensión así como el valor de los parámetros de los modelos linealizados de los dispositivos semiconductores y de las fuentes controladas no lineales.

.LIB <"nombre de fichero">*: ficheros donde se deberán buscar los modelos y los subcircuitos.

.INC <"nombre de fichero"> : inserta el fichero especificado en el fichero actual. No puede contener la instrucción ".END".

.TEXT <<nombre>=<"cadena de caracteres">*: reemplaza nombre por la cadena de caracteres. <nombre> no puede ser el nombre de un ".PARAM".

.WATCH [DC] [AC] [TRAN] [<variable> [<límite inferior>,<límite superior>]]*: visualizará el valor de la variable al realizarse el análisis indicado. Si se indican límites el análisis se parará en el caso de que la variable traspase alguno de los límites; a partir de aquí se puede abortar el análisis o bien continuar.

.WIDTH OUT=<valor> : igual que WIDTH de ".OPTIONS".

`.OPTIONS [nombre de la opción]* [<nombre de la opción>=<valor>]`: mediante esta instrucción se fijan diversas opciones de simulación. Existen dos tipos de opciones:

1) *opciones tipo flag (activadas/desactivadas):*

En estas opciones basta poner el nombre de la opción para que se activen. Las opciones más significativas de este tipo son:

- NOBIAS: suprime el listado de las tensiones del punto de reposo
- NODE: proporciona un listado completo de las conexiones
- NOREUSE: no salva ni recupera automáticamente la información del punto de reposo en análisis multi-ejecución (STEP, TEMP), Motecarlo y "caso peor" (worts case)
- OPTS: proporciona un listado con todas las opciones
- WIDTH: fija el número de columnas de la salida. Pueden ser 80 (por defecto) ó 132

Ejemplo:

`.OPTIONS NODE`

2) *opciones que necesitan un valor*

su sintaxis es:

`".OPTIONS <nombre de la opción>=<value>"`

Las opciones más significativas de este tipo son:

- DEFAD=..., DEFAS=...: fijan el área por defecto de los drenadores y surtidores de los MOSFET (por defecto: 0)
- DEFL=..., DEFW=...: indican la longitud y la anchura por defecto de los MOSFET (por defecto: 100 μ)
- TNOM=...: temperatura por defecto (por defecto: 27 °C)
- GMIN=...: conductancia mínima para cualquier rama (por defecto: 1E-12 Ω^{-1})
- ABSTOL=...: error permitido en el cálculo de corrientes (por defecto: 1pA)
- VNTOL=...: error permitido en el cálculo de tensiones (por defecto: 1 μ V)
- CHGTOL=...: error permitido en el cálculo de cargas (por defecto: 0,1pC)
- RETOL=...: error relativo para corrientes y tensiones (por defecto: 0,001)

1.6 Entorno integrado (*Control Shell*)

El entorno integrado (o *Shell* de control) es un programa que se encarga de ejecutar los diversos programas que componen PSpice y que cuenta además con la capacidad de modificar el fichero fuente a requerimiento del usuario. Todo ello se realiza sobre un entorno de menús que facilita de esta forma el manejo del programa. No se hará aquí una descripción exhaustiva de este entorno ya que, por un lado, aún no se dispone de suficientes conocimientos para comprender el significado de todas las posibilidades y, por otro lado, los propios menús ilustran sobre las posibilidades del entorno. Así pues, por ahora se explicará tan sólo la manera de activarlo y las ideas básicas. Otras posibilidades se comentarán cuando se estudien los conceptos con los que está relacionados, y quedarán el resto de opciones para que el lector las experimente.

Al entorno integrado se accede con la orden *PS*. Al ejecutar este programa aparecerá una pantalla de menús pero en el que sólo tres de ellos están realizados: *Files*, *Probe*, *Quit*. *Probe* se estudiará más adelante (capítulo 5) y *Quit* se utilizará para terminar la sesión o para ejecutar un comando del DOS.

El menú *Files* dispone de las siguientes opciones:

- Edit
- Browse Output
- Current File
- Save File
- X-External Editor
- R-External Browser
- Display/Prn Setup

La opción "Current File" selecciona el fichero de circuito con el que se va a trabajar. Si el fichero ya existía se cargará y en la línea inferior aparecerá el nombre del fichero con la palabra "Loaded". Si, por el contrario, se trata de un fichero nuevo se generará, hecho que se indica con la palabra "New". Todas las acciones del entorno integrado se realizarán usando el nombre de este fichero.

A continuación se puede editar el fichero con la opción "Edit". La opción "Browse Output" sirve para examinar el fichero de salida ".OUT".

"Save File" guarda el fichero fuente en disco. Mientras no se ejecute esta opción el fichero estará sólo en memoria.

"X-External Editor" y "R-External Browser" permiten utilizar un editor diferente del que incorpora el entorno integrado. El editor al cual se llama se define a través de la variable de sistema PSEDIT. Si además se desea que a este editor se le pase como argumento el nombre del fichero fuente que se está utilizando se debe añadir: "%f". Por ejemplo para utilizar el editor del MS-DOS pasando automáticamente el nombre del fichero fuente se ejecutará la siguiente orden del DOS para definir la variable PSEDIT:

```
SET PSEDIT = c:\DOS\EDIT %f
```

Si se desea disponer de este editor en sucesivas sesiones esta orden se deberá añadir en el fichero "autoexec.bat".

Por último la opción "Display/Prn Setup" configura la pantalla y el dispositivo de impresión. Es equivalente, pero más fácil de usar, a ejecutar el programa SETUPDEV que se describe en el capítulo 5.

Una vez que se ha seleccionado el fichero de circuito –"Current File"– las opciones que no estaban realizadas pasan a estarlo, indicando de esta forma que se puede acceder a ellas.

Algunas opciones son capaces de añadir líneas al fichero de circuito. Una de ellas es "Change Options" del menú "Analysis" que permite ver el estado de las opciones que se comentaron en la sección anterior y modificarlas. Por ejemplo para cambiar la opción "Library", que por defecto es "N", a "Y" hay que situarse sobre la opción y cambiar a "Y". Para que el cambio tenga efecto se han de pulsar simultáneamente las teclas "Control" + "Return". Si después de hacer esto se vuelve a editar el fichero fuente se observará que se ha añadido la siguiente línea:

```
.options library ;*ipsp*
```

Esta línea ha sido añadida por el Shell de Control. Cuando una acción del entorno integrado añade alguna línea lo indica poniendo **ipsp** al final de la línea.

2 Tipos de análisis

En SPICE se pueden realizar diversos tipos de análisis sobre el circuito descrito. Cada uno de ellos se especifica mediante la correspondiente línea de comandos, que se compondrá de una identificación del tipo de análisis que se debe realizar y del entorno en que éste se efectuará. El entorno difiere de un tipo de análisis a otro ya que es propio de cada uno. Se explicará la sintaxis de cada análisis tal como se introduciría "a mano", pero téngase en cuenta que se puede usar *el entorno integrado* para introducir los análisis. Esto se hace a través del menú *Analysis*.

Los análisis que se estudiarán son:

- a) Barrido en continua: ".DC"
- b) Respuesta en frecuencia: ".AC"
- c) Transitorio o temporal: ".TRAN"
- d) Función de transferencia: ".TF"
- e) Análisis de Sensibilidad: ".SENS"
- f) Análisis de Fourier: ".FOUR"

2.1 DC (Barrido de continua)

Este análisis ya es conocido dado que se usó en el capítulo anterior. Como se recordará un análisis de este tipo asigna un valor a una variable, calcula la respuesta en continua, a continuación incrementa la variable en un valor especificado y vuelve a calcular la respuesta. El proceso continúa hasta alcanzar el valor final.

Se llama análisis en DC pues lo que se hace es un cálculo del punto de reposo con cada uno de los valores de la variable especificada sin tener en cuenta el tiempo. Esto significa que un condensador será tratado como un circuito abierto, las inductancias como cortocircuitos y se utilizarán los valores de continua de las fuentes independientes¹ a no ser que la variable de barrido sea precisamente el valor

¹ El valor de continua (valor del parámetro DC) de una fuente de tensión se comentó en el capítulo anterior. Lo mismo es aplicable a una fuente de corriente. El estudio de estas fuentes se cubre en el capítulo 3

de una fuente independiente en cuyo caso el valor ordenado en el análisis toma preferencia sobre el fijado al definir el componente.

La sintaxis más sencilla de la instrucción que inicia un análisis DC es la siguiente:

.DC <variable> <valor inicial> <valor final> <paso>

<variable>: es la magnitud que se variará. No tiene por qué ser un valor de continua (DC) de una fuente independiente sino que puede ser cualquiera de los siguientes tipos de variables:

- a) Valor DC de una fuente independiente: *<variable>* debe ser el nombre de una fuente independiente.
- b) Parámetros globales: se trató de ellos en el capítulo 1. En este caso *<variable>* se compondrá de la palabra clave PARAM seguida del parámetro global que se desea barrer. Ex: *.DC PARAM CONST . . .*
- c) Parámetros de modelos de dispositivos: son otros parámetros, diferentes a los globales y que se utilizan para particularizar un determinado dispositivo. Se estudiarán en el capítulo 3. En este caso *<variable>* contiene el nombre del modelo y el parámetro que se debe modificar.
- d) Temperatura: permite analizar el circuito a diferentes temperaturas. Los dispositivos disponen de unos parámetros que marcan su dependencia con la temperatura por lo que se puede estudiar el comportamiento del circuito en función de la temperatura. Para ello *<variable>* debe ser la palabra "TEMP".

<valor inicial> <valor final>: indican el rango de valores que tomará la variable de barrido. *<valor inicial>* puede ser mayor o menor que el *<valor final>*.

<paso>: con este parámetro se determina el incremento que se aplicará a la variable de barrido. El sentido del barrido depende de si *<valor inicial>* es mayor que *<valor final>* o a la inversa y no del signo de *<paso>* que debe ser siempre una cantidad positiva.

Con la sentencia anterior se efectuará un barrido lineal –éste es el tipo de barrido que se toma por defecto– pero pueden realizarse también barridos logarítmicos o incluso se puede proporcionar una lista de valores. Debido a estas posibilidades la sintaxis es más compleja que la mostrada anteriormente. Se dispone de las siguientes posibilidades:

Barrido lineal

.DC [LIN] <variable> <valor inicial> <valor final> <paso>

Barrido logarítmico

.DC {DEC/OCT} <variable> <valor inicial> <valor final> <puntos>

Lista de valores

`.DC <variable> LIST <lista de valores>`

El tipo de barrido por defecto, el que se realizará si no se especifica ningún tipo de barrido, es el lineal "LIN". Pero también se puede hacer un barrido logarítmico, por décadas o por octavas, con las opciones "DEC" o "OCT". En estos casos < puntos > será la cantidad de valores por década (si se indicó "DEC") o por octava (si se indicó "OCT").

En el barrido de lista de valores se ha de proporcionar la lista de valores separados entre sí por un espacio en blanco.




Fig. 2.1

Por ejemplo, si sobre el circuito de la figura 2.1 se desea un barrido logarítmico, por décadas, de la fuente de corriente "IP" entre $1\mu\text{A}$ y 10mA , con seis puntos de análisis por cada década, el fichero de entrada sería:

```
EX2_1 (Barrido en continua logarítmico)
```

```
VT 1 0 2V
```

```
R1 1 22 2K
```

```
IP 2 22 1M
```

```
R2 1 3 4K
```

```
R3 2 0 2K
```

```
R4 3 0 4K
```

```
.DC DEC IP 1U 10M 6
```

```
.PRINT DC V(1,2)
```

```
.PROBE
```

```
.END
```

La visualización de los resultados mediante PROBE es la mostrada en la figura 2.2. Se obtiene una curva en la que el valor de X es la variable que se varía y en el eje Y se visualiza la traza seleccionada.




Fig. 2.2

El resultado de la instrucción .PRINT se guarda en el fichero ".OUT" en forma de una tabla de valores. El fragmento de ".OUT" que refleja este análisis es:

```

****  DC TRANSFER CURVES          TEMPERATURE = 27.000 DEG C
*****
      IP          V(1,2)
1.000E-06  2.002E+00
1.468E-06  2.003E+00
2.154E-06  2.004E+00
3.162E-06  2.006E+00
4.642E-06  2.009E+00
6.813E-06  2.014E+00
1.000E-05  2.020E+00
1.468E-05  2.029E+00
2.154E-05  2.043E+00
3.162E-05  2.063E+00
4.642E-05  2.093E+00

```

6.813E-05	2.136E+00
1.000E-04	2.200E+00
1.468E-04	2.294E+00
2.154E-04	2.431E+00
3.162E-04	2.632E+00
4.642E-04	2.928E+00
6.813E-04	3.363E+00
1.000E-03	4.000E+00
1.468E-03	4.936E+00
2.154E-03	6.309E+00
3.162E-03	8.325E+00
4.642E-03	1.128E+01
6.813E-03	1.563E+01
1.000E-02	2.200E+01

En fondo sombreado se ha marcado una década completa. Como puede observarse se han dado seis valores por década a la corriente IP, tal como se indicó en la instrucción *.DC*.

Un último aspecto que debe considerarse es la posibilidad de realizar barridos anidados. Con un barrido anidado se puede establecer un barrido adicional sobre una segunda variable de forma tal que, para cada valor de la segunda variable, se hace un barrido de la primera variable. La segunda variable y el tipo de barrido sobre esta segunda variable se indican en la misma sentencia *.DC* a continuación de la especificación del primer barrido. Ambos barridos no tienen por qué ser iguales, uno puede ser logarítmico y el otro lineal, etc. Una posible sintaxis es :

```
.DC [DEC|OCT] <variable> <valor inicial> <valor final> <puntos>
+           [especificación del barrido anidado]
```

La visualización de los resultados mediante PROBE proporciona una familia de curvas y cada una de ellas corresponde a un valor de la primera variable.

Por ejemplo, se puede utilizar el barrido anidado para obtener las curvas de salida de un transistor bipolar. Para ello se utilizará el circuito de la figura 2.3. En este circuito se introduce un nuevo elemento, el transistor bipolar. La forma de definir un transistor bipolar, y en general todo semiconductor, difiere de las definiciones de los componentes vistos hasta ahora. Esto es así porque para caracterizar un transistor se necesitan muchos más parámetros que para, por ejemplo, una resistencia. Esta hecho haría muy tediosa la introducción de un dispositivo de este tipo. Para simplificar esta tarea se agrupan todos los parámetros de un tipo de transistor determinado bajo lo que se llama un *MODELO*. La definición de los modelos se realiza mediante la instrucción *.MODEL*.²

² Aunque no se han utilizado las resistencias, los condensadores, etc. también admiten modelos pero habitualmente no se utilizan. Al estudiar estos componentes se verán sus modelos.




Fig. 2.3

Por ejemplo, el transistor bipolar BC-108B queda descrito de la siguiente forma:

```
.MODEL BC108B NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=59.59 Bf=381.7 Ise=59.74f
+      Ne=1.522 Ikf=3.289 Nk=.5 Xtb=1.5 Br=2.359 Isc=192.9p Nc=1.954
+      Ikr=7.807 Rc=1.427 Cjc=5.38p Mjc=.329 Vjc=.6218 Fc=.5 Cje=11.5p
+      Mje=.2718 Vje=.5 Tr=10n Tf=438p Itf=5.716 Xtf=14.51 Vtf=10)
```

Con estas líneas se ha definido el modelo de nombre "BC108B" que corresponde a las características de un transistor bipolar NPN (indicado por la palabra NPN).³

Al utilizar un transistor bipolar en un circuito lo primero que se ha de indicar es que se trata de este tipo de componente. Para ello el nombre del componente debe comenzar con una "Q". Después se proporcionan los nodos de conexión (colector, base, emisor) y por último el tipo de transistor (nombre del modelo).

Por ejemplo, si el transistor que se va a simular es un BC-108B con el nombre de "TEST" y tiene el colector conectado al nodo 2, la base al nodo 1 y el emisor al nodo 0, la línea de declaración será:

```
QTEST 2 1 0 BC108B
```

Volviendo al análisis ".DC" anidado si se desean obtener las curvas características de salida se proporcionará un corriente constante a la base del transistor y se hará un barrido de la tensión V_{CE} . Esto se consigue utilizando un análisis DC con anidamiento. La primera variable debe ser la tensión V_{CE} y la segunda variable será la corriente I_B . El circuito será:

```
EX2_2 (Curvas del transistor BC-108B)
```

```
.MODEL BC108B NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=59.59 Bf=381.7 Ise=59.74f
+      Ne=1.522 Ikf=3.289 Xtb=1.5 Br=2.359 Isc=192.9p Nc=1.954
```

³ La forma de definir modelos y los parámetros de cada modelo se verá en los capítulos 3 y 4

```

+      Ikr=7.807 Rc=1.427 Cjc=5.38p Mjc=.329 Vjc=.6218 Fc=.5 Cje=11.5p
+      Mje=.2718 Vje=.5 Tr=10n Tf=438p Itf=5.716 Xtf=14.51 Vtf=10)

```

```
Q1 2 1 0 BC108B
```

```
VCE 2 0 5V
```

```
IB 0 1 1UA
```

```
.DC DEC VCE .05 5 10 LIN IB .01M .1M .01M
```

```
.PROBE
```

```
.END
```

Se efectúa un barrido logarítmico de la tensión V_{CE} entre 0.05V y 5V con 10 valores en cada década para cada valor de la corriente I_B comprendido entre 0.01mA y 0.1mA con incrementos lineales de 0.01mA. Como puede verse se han combinado dos tipos de barrido diferentes, uno logarítmico y uno lineal.




Fig. 2.4

El resultado utilizando PROBE proporciona la gráfica de la figura 2.2 que, como se observa corresponde a la corriente de colector en relación a la tensión V_{CE} y I_B . Las referencias de corriente se toman entrantes en todos los terminales del transistor. Nótese que aunque se ha realizado un barrido

logarítmico de la tensión V_{CE} en la representación el eje es lineal. Realmente los puntos calculados están repartidos logarítmicamente, pero el tipo de eje sobre el que se representará se define en el programa gráfico tal como se verá en el capítulo dedicado a PROBE.

2.2 AC (Respuesta en frecuencia)

Con este análisis se puede obtener la respuesta del circuito a señales senoidales de diversas frecuencias. La sintaxis es la siguiente:

`.AC {LIN/DEC/OCT} <puntos> <frecuencia inicial> <frecuencia final>`

donde:

`{LIN/DEC/OCT}`: se ha de especificar una de las tres opciones. Indica el modo en que se realizará el barrido. Este podrá ser lineal "LIN", logarítmico por décadas "DEC" o por octavas "OCT".

`<Puntos>`: fija el número total de frecuencias en el caso de variación lineal,⁴ o el número de frecuencias por década si el barrido es logarítmico.

`<frecuencia inicial> <frecuencia final>`: rango de frecuencias entre los que se efectuará el análisis.

Las señales a las frecuencias indicadas se aplican simultáneamente a todos aquellos generadores independientes que tienen una especificación AC. En el capítulo anterior se introdujo la sintaxis de una fuente independiente de tensión, en ella aparece la palabra AC entre corchetes. Si en la declaración




Fig. 2.5

⁴ Nótese que difiere del significado que tenía en el análisis DC lineal.

de la fuente aparece esta palabra se establece que el generador se comportará como un generador senoidal con la amplitud y, opcionalmente, con la fase indicada. La frecuencia de la señal la establece el análisis ".AC". Consúltese el capítulo siguiente para una descripción completa de los generadores independientes.

A modo de ejemplo se analizará la respuesta en frecuencia del circuito de la figura 2.2.

El fichero de circuito será:

```

EX2_3 (Respuesta en frecuencia)

.model BC108B  NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=59.59 Bf=381.7 Ise=59.74f
+      Ne=1.522 Ikf=3.289 Xtb=1.5 Br=2.359 Isc=192.9p Nc=1.954
+      Ikr=7.807 Rc=1.427 Cjc=5.38p Mjc=.329 Vjc=.6218 Fc=.5 Cje=11.5p
+      Mje=.2718 Vje=.5 Tr=10n Tf=438p Itf=5.716 Xtf=14.51 Vtf=10)

VIN 1 0 AC 1E-3 ; generador de señal
VCC 4 0 6V      ; alimentación

RB 1 5 50
CIN 5 2 10U

R1 4 2 390K
R2 2 0 195K
RC 4 3 1K
Q1 3 2 0 BC108B

.AC DEC 10 10 5MEG
.PROBE

.END

```

Con esto se analizará la respuesta del circuito para frecuencias del generador VIN entre 10Hz y 5MHz tal como se estipula en las líneas que aparecen sombreadas. El resultado es el mostrado en la figura 2.6.

Un aspecto importante que se debe remarcar es que el análisis ".AC" se realiza usando el modelo linealizado en pequeña señal. Los dispositivos no lineales como diodos, BJT, FET, etc., disponen de diversos modelos de funcionamiento que se obtienen a partir de los datos suministrados con el modelo SPICE (el definido con .MODEL). Uno de ellos es el modelo en pequeña señal que describe un funcionamiento lineal alrededor del punto de reposo y que sólo es válido para pequeñas variaciones de dicho punto. Esta es la misma técnica que se utiliza en un análisis "manual".




Fig. 2.6

Los valores del modelo en pequeña señal dependen del transistor y del punto de reposo. El uso del modelo en pequeña señal permite realizar un cálculo rápido por lo que se puede estudiar el circuito a muchas frecuencias, lo que resultaría imposible en el caso de utilizar el modelo no lineal. El punto de reposo se calcula considerando los valores DC de las fuentes independientes.

Hay que tener siempre en cuenta este hecho ya que en caso contrario se puede llegar a resultados erróneos. Por ejemplo, si supusiéramos que la excitación es de 1V en lugar de 1mV los resultados serían los mostrados en la figura 2.7. Según estos resultados la tensión de colector a 1KHz sería de unos 100V, lo cual es a todas luces absurdo considerando las tensiones de alimentación del circuito. En realidad el transistor entrará en saturación, pero esto no se puede detectar en este análisis ya que se asume un modelo lineal para describir el funcionamiento del transistor.

Puesto que se supone que se está trabajando con un circuito lineal la magnitud de la excitación no tiene importancia. En muchas ocasiones lo que interesa son ganancias, esto es, relaciones entre dos magnitudes, por lo que es útil dar a la excitación un valor que facilite los cálculos, por ejemplo 1V.

Las figuras 2.8 y 2.9 ilustran este hecho. La primera corresponde al *Bode* de la ganancia entre la tensión de salida (colector) y la tensión de entrada (V(1)) cuando la señal de entrada es de 1mV. La segunda corresponde al mismo *Bode* pero para una señal de entrada de 1V. Se observa que los resultados son iguales.




Fig. 2.7




Fig. 2.8




Fig. 2.9

Nota: Obsérvese que, en cada caso, se ha definido la traza de un modo diferente. La expresión de la traza que se está visualizando figura al pie de la gráfica precedida por un cuadrado. En el primer caso la traza es una expresión entre variables del circuito que proporciona la ganancia en dB. El mismo resultado se puede obtener, de forma más directa, usando funciones específicas disponibles en el programa PROBE (en el capítulo dedicado a este programa se recogen las funciones permitidas). Una de ellas es "DB(x)" que proporciona el valor de 'x' en dB. Esta es la función que se ha empleado en la segunda gráfica, equivalente a la expresión de la gráfica de la figura 2.8.

En el análisis ".AC" las variables son cantidades complejas y por ello se necesita especificar si se desea visualizar la magnitud –que se toma por defecto–, la fase, o bien la parte real o la parte imaginaria. Ello se indica a través de un prefijo tal como puede verse en la definición de la traza de la figura 2.10, en la que se ha usado el prefijo "P" para especificar que se desea visualizar la fase. Para la parte real se utiliza el prefijo "R" y para la imaginaria el prefijo "IMG".

Pero para conocer el comportamiento real de este circuito se necesita observar los márgenes de cada una de las zonas de funcionamiento lo que obliga a considerar la característica no lineal del transistor.

Para estudiar este tema, así como el funcionamiento fuera del régimen permanente, se dispone del análisis transitorio.




Fig. 2.10

2.3 TRAN (Transitorio)

El objetivo del análisis ".TRAN" es analizar el comportamiento del circuito al transcurrir el tiempo, sea en régimen transitorio o permanente, por lo que se le puede considerar como un análisis temporal.

Con el análisis transitorio se estudia la respuesta del circuito a una determinada excitación o estímulo a lo largo del tiempo. El formato más simple de la instrucción es:

.TRAN[/OP] <intervalos de representación> <tiempo final>

Donde:

[/OP]: es opcional. Proporcionará información detallada de los valores iniciales, tal como se explica en el apartado *punto de reposo* de esta sección.

<intervalos de representación>: determina los instantes de tiempo para los que se representarán los resultados. Estos tiempos no coinciden con los tiempos para los que se ha hecho el análisis, por ello se realiza una interpolación para obtener los valores de representación.

<tiempo final>: tiempo en el que se terminará el análisis.

Se efectuará un análisis desde el tiempo igual a cero hasta que se alcance el tiempo final. Los intervalos de tiempo entre los que se realiza el análisis vienen determinados internamente. Si la variación es muy grande los incrementos de tiempo se hacen pequeños; por el contrario, si la variación es pequeña los incrementos son más grandes. Hay un límite para el máximo incremento que es: <tiempo final>/50.

La excitación o estímulo es la proporcionada por aquellas fuentes independientes en las que se usó la opción [transitorio].⁵ El estímulo es una de las siguientes formas de onda recogidas en el apéndice D:

- Senoidal: define una senoide. Se puede especificar *offset*, amplitud, fase, frecuencia, etc. Para declarar un estímulo de este tipo [transitorio] toma la forma: *SIN*(<parámetros>).
- Exponencial: proporciona una exponencial creciente y a continuación una decreciente. Se puede fijar una constante de tiempo creciente y otra decreciente. Es una señal no periódica. Palabra clave: *EXP*(<parámetros>).
- Pulso: define un pulso, es decir, una señal que toma dos valores cada uno de ellos durante un tiempo determinado y se repite (por tanto es periódica). Palabra clave: *PULSE*(<parámetros>).
- Sintetizada: permite "dibujar" cualquier forma de onda si se suministran los puntos de la señal. Palabra clave: *PWL*(<parámetros>).
- Senoide modulada en frecuencia: genera una portadora modulada en frecuencia por un tono de frecuencia fija. Palabra clave: *SFFM*(<parámetros>).

Estos estímulos se pueden incluir haciendo uso del entorno de usuario (*Control Shell*). El menú *StmEd* llama al programa *Estimulus Editor* que permite incluir generadores independientes con estímulos de transitorios. Al arrancar este programa aparece una pantalla y una serie de opciones y se puede definir un nuevo estímulo o editar uno de los ya existentes. No obstante la versión educacional sólo permite definir señales de tipo senoidal desde el *Estimulus Editor*.

En el análisis se utiliza el llamado *modelo en gran señal* de los dispositivos no lineales. Este modelo no está restringido a pequeñas variaciones alrededor del punto de reposo, por tanto da una idea precisa del comportamiento del circuito.

⁵ Véanse las sentencias de declaración de fuentes independientes en el apéndice C.

Con el siguiente fichero de circuito se hace un análisis transitorio sobre el mismo circuito que se utilizó en el análisis ".AC":

```

EX2_4A ANALISIS TRANSITORIO DE UN EMISOR COMUN

.model BC108B NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=59.59 Bf=381.7 Ise=59.74f
+      Ne=1.522 Ikf=3.289 Xtb=1.5 Br=2.359 Isc=192.9p Nc=1.954
+      Ikr=7.807 Rc=1.427 Cjc=5.38p Mjc=.329 Vjc=.6218 Fc=.5 Cje=11.5p
+      Mje=.2718 Vje=.5 Tr=10n Tf=438p Itf=5.716 Xtf=14.51 Vtf=10)

VIN 1 0 AC 1E-3 SIN(0 0.5E-3 1K); generador de señal
VCC 4 0 6V          ; alimentación

RB 1 5 50
CIN 5 2 10U

R1 4 2 390K
R2 2 0 195K
RC 4 3 1K
Q1 3 2 0 BC108B;Transistor

.TRAN 1U 4M
.PROBE

.END

```

En el generador de señal se ha especificado el tipo de estímulo como una señal senoidal con 0V de *offset*, amplitud de 0,5mV y frecuencia de 1KHz. En la sentencia del análisis transitorio se ha solicitado un paso de visualización de 1µs (1U) y un tiempo final de 4ms (4M). El resultado es el mostrado en la gráfica de la figura 2.11.

Según estas gráficas la señal de salida, VC(Q1), tiene una amplitud pico a pico aproximada de 107mV (obsérvense los valores de los cursores "C1" y "C2" y la diferencia "dif" que aparecen en el recuadro – la primera columna indica el tiempo y la segunda el valor de la variable–, mientras que la amplitud de la señal de entrada, V(1) es de 1mV pico a pico. En consecuencia, la ganancia es de 107. Este resultado es acorde con el obtenido en el análisis ".AC" de este mismo circuito a 1KHz.

Si se aumenta la excitación de forma que pase a valer 1V el resultado será el de la figura 2.12. En este caso se tiene en cuenta el hecho de que el transistor no estará siempre en la zona activa, al contrario que en el análisis ".AC", donde se consideraba siempre el transistor en la zona lineal y se obtenían por tanto resultados falsos.




Fig. 2.11




Fig. 2.12

Punto de reposo

El punto de reposo y las condiciones iniciales son dos temas relacionados. Se calculan dos puntos de reposo diferentes. Uno –el tratado hasta ahora– se aplica a todos los análisis excepto al ".TRAN". Es el que se muestra bajo la sección "SMALL SIGNAL BIAS SOLUTION" del fichero ".OUT". En el cálculo de este punto de reposo se usa las especificaciones DC de las fuentes independientes y las condiciones iniciales fijadas mediante la sentencia ".IC" que se verá seguidamente. Se calculan las tensiones de cada nodo, las corrientes a través de las fuentes independientes y los valores de los modelos linealizados.⁶ Esta es la información que se suministra al incluir la sentencia ".OP".

Además, si se ha indicado un análisis ".TRAN" se calculará otro punto de reposo –llamado *valor inicial*– que se mostrará en la sección "INITIAL TRANSIENT SOLUTION". En este caso se trata del valor correspondiente al tiempo cero, es decir, al inicio del análisis transitorio. Para calcularlo se utiliza, bien el valor DC de las fuentes independientes bien, si éste no existe, el valor inicial del estímulo. Se utilizan las condiciones iniciales indicadas con ".IC" a no ser que se utilice la opción "UIC", como se comentará posteriormente. Igual que en el caso anterior se calculan las tensiones y las corrientes y *de nuevo* los valores de los modelos linealizados, aunque estos modelos no se utilizan en el análisis transitorio. Para obtener información detallada se ha de incluir la palabra "/OP" en la especificación del análisis transitorio.

Un ejemplo puede ayudar a comprender las diferencias entre ambos puntos de reposo. El siguiente fichero de circuito describe el circuito de la figura 2.5 pero sin la resistencia R2 ni el condensador CIN. Se ha especificado un análisis transitorio y un análisis ".AC". Además el generador VIN contiene especificación DC y estímulo de transitorio. Obsérvese que el valor inicial del estímulo "2" no coincide con el valor de DC que es cero. Debido a esto último el punto de reposo para el análisis ".AC" diferirá del punto de reposo o valor inicial del transitorio. SPICE considerará que el generador VIN tiene un valor de 0V para el cálculo del punto de reposo del análisis ".AC" y de 2V para el cálculo del punto de reposo del transitorio.

Estado inicial	
.model	BC108B NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=59.59 Bf=381.7 Ise=59.74f
+	Ne=1.522 Ikf=3.289 Nk=.5 Xtb=1.5 Br=2.359 Isc=192.9p Nc=1.954
+	Ikr=7.807 Rc=1.427 Cjc=5.38p Mjc=.329 Vjc=.6218 Fc=.5 Cje=11.5p
+	Mje=.2718 Vje=.5 Tr=10n Tf=438p Itf=5.716 Xtf=14.51 Vtf=10)
*	PHILIPS pid=bc107b case=TO18
*	91-08-02 dsq
VIN 1 0 DC 0 AC 1E-3 SIN(2 0.5E-3 1K)	
VCC 4 0 6V	

⁶ Véanse en el capítulo 4 los dispositivos que disponen de modelo linealizado.

```

RB 1 2 195K
R1 4 2 390K
RC 4 3 1K
Q1 3 2 0 BC108B

.TRAN/OP 1U 4M
.AC DEC 10 10 1.000k
.OP

.END

```

Como se ha incluido la sentencia ":OP" y la opción "/OP" se obtiene la siguiente información en el fichero ".OUT":

```

**** SMALL SIGNAL BIAS SOLUTION   TEMPERATURE = 27.000 DEG C
*****
NODE VOLTAGE  NODE VOLTAGE  NODE VOLTAGE  NODE VOLTAGE
( 1) 0.0000 ( 2) .6914 ( 3) 3.0174 ( 4) 6.0000

VOLTAGE SOURCE CURRENTS
NAME      CURRENT
VIN       3.546E-06
VCC       -2.996E-03

TOTAL POWER DISSIPATION 1.80E-02 WATTS

**** OPERATING POINT INFORMATION   TEMPERATURE = 27.000 DEG C
*****
**** BIPOLAR JUNCTION TRANSISTORS

NAME      Q1
MODEL     BC108B
IB        1.01E-05
IC        2.98E-03
VBE       6.91E-01
VBC       -2.33E+00
VCE       3.02E+00

```

```

BETADC  2.96E+02
GM      1.15E-01
RPI     2.81E+03
RX      0.00E+00
RO      2.08E+04
CBE     7.10E-11
CBC     3.23E-12
CBX     0.00E+00
CJS     0.00E+00
BETAAC  3.24E+02
FT      2.47E+08

```

```

•
•
•

```

```

**** INITIAL TRANSIENT SOLUTION TEMPERATURE = 27.000 DEG C

```

```

*****
NODE VOLTAGE  NODE VOLTAGE  NODE VOLTAGE  NODE VOLTAGE
( 1) 2.0000 ( 2) .7095 ( 3) .2851 ( 4) 6.0000

```

VOLTAGE SOURCE CURRENTS

```

NAME      CURRENT

```

```

VIN      -6.618E-06

```

```

VCC      -5.728E-03

```

```

TOTAL POWER DISSIPATION 3.44E-02 WATTS

```

```

**** OPERATING POINT INFORMATION TEMPERATURE = 27.000 DEG C

```

```

*****

```

**** BIPOLAR JUNCTION TRANSISTORS

```

NAME      Q1
MODEL     BC108B
IB        2.02E-05
IC        5.71E-03
VBE       7.09E-01
VBC       4.24E-01
VCE       2.85E-01

```

BETADC	2.83E+02
GM	2.21E-01
RPI	1.46E+03
RX	0.00E+00
RO	9.84E+03
CBE	1.17E-10
CBC	7.68E-12
CBX	0.00E+00
CJS	0.00E+00
BETAAC	3.21E+02
FT	2.80E+08

Se puede comprobar que las secciones "node voltage ...", "voltage source currents" y "operating point information" están duplicadas. La primera vez se corresponden con el punto de reposo del análisis AC: SMALL SIGNAL BIAS SOLUTION y la segunda con el análisis transitorio: *INITIAL TRANSIENT SOLUTION*.

En este ejemplo se ve que los valores son diferentes. Además, mirando la tensión colector-emisor, V(3), se ve que en el primer caso el transistor está en activa y en el segundo en saturación. Esta diferencia también se aprecia en los valores asignados para el modelo linealizado, que son diferentes en uno y otro caso.

Las condiciones iniciales se pueden fijar de las siguientes maneras:

a) Comando ".IC"

Mediante este comando se puede fijar la tensión en uno o más nodos del circuito. Por ejemplo:

```
.IC V(2)=5 V(3,4)=2.8
```

pone inicialmente el nodo 2 a 5V y fija una tensión entre los nodos 3 y 4 de 2,8V. Esta instrucción afecta al cálculo del punto de reposo de todos los análisis. El efecto de esta instrucción es el siguiente: para el cálculo del punto de reposo se conecta en cada nodo especificado una fuente de tensión con una resistencia en serie de valor $0,002\Omega$.⁷ Una vez completado el cálculo del punto de reposo se desactivan las fuentes de tensión y comienza el análisis.

b) En la definición de componentes "IC=..."

Las condiciones iniciales de los condensadores (tensión inicial) y de las inductancias (corriente inicial) se pueden fijar utilizando la opción "IC=..." en la definición de estos componentes. Pero a diferencia del caso anterior esta información sólo se usa para el cál-

⁷ La inclusión de la resistencia en serie se realiza para prevenir bucles de tensión. Normalmente no produce ningún otro efecto, pero pueden existir situaciones (por ejemplo en un circuito integrador) en que esta resistencia afecte significativamente al resultado y no se obtenga la tensión deseada en el nodo.

culo del punto de reposo del análisis transitorio y no para los de pequeña señal, siempre y cuando en la especificación del análisis transitorio se haya incluido la opción "UIC" (*Use Initial Conditions*). De hecho al usar UIC se omite el cálculo del punto de reposo y el análisis comienza inmediatamente haciendo uso de los valores fijados mediante "IC=...", por lo que las sentencias ".IC" no surten efecto en este caso. Tanto es así que si se usa UIC no aparecerá la sección *INITIAL TRANSIENT SOLUTION*.

Por ejemplo, si se desea observar la carga del condensador del circuito de la figura 2.13, cuando la tensión inicial en el condensador es de 2V se puede utilizar el siguiente fichero:

```

EX2_5 Carga y descarga de un condensador

VIP 1 0 PULSE( 0 6 0 1U 1U 10M 20M) ;pulso de 6 voltios de pico, periodo
                                ;de 20ms y ciclo de trabajo 50%

R1 1 2 1K
C1 2 0 1U
.IC V(2)=2V                      ;Condición inicial V condensador: 2V

.TRAN 1U 20M
.PROBE

.END

```

El resultado es el que se observa en la figura 2.14 (V(1) es el pulso de entrada y V(2) la tensión del condensador).




Fig. 2.13




Fig. 2.14

Si se opta por el segundo método de fijar las condiciones iniciales el circuito será:

```

EX2_5B Carga y descarga de un condensador.

VIP 1 0 PULSE( 0 6 0 1U 1U 10M 20M) ;pulso de 6 voltios de pico, periodo
;de 20ms y ciclo de trabajo 50%

R1 1 2 1K
C1 2 0 1U IC=2

.TRAN 1U 20M UIC
.PROBE

.END

```

En ambos casos se obtienen los mismos resultados.

El tiempo inicial, a partir del cual se comienza el análisis, siempre es cero. Pero se puede instruir al programa para que sólo "imprima" los resultados a partir de un instante determinado. Para ello se utiliza el parámetro opcional [*no print value*].

Un último parámetro opcional, disponible en este análisis, es [*step ceiling value*] (límite de incremento de tiempo) que permite modificar el límite del máximo incremento de tiempo que por defecto es : <tiempo final>/50.

En resumen, la sintaxis completa del análisis transitorio es:

```
.TRAN[/OP] <intervalos de representación> <tiempo final>
+      [no print value] [step ceiling value] [UIC]
```

2.4 TF (Función de transferencia)

Este análisis proporciona información sobre la ganancia entre dos variables de circuito (o la transconductancia o transresistencia), impedancia de entrada e impedancia de salida.. Su sintaxis es:

```
.TF <variable de salida> <fuente indep. de entrada>
```

La variable de entrada tiene que ser el nombre de una fuente independiente de tensión o corriente. La variable de salida tiene el mismo formato y significado que en la sentencia ".PRINT", pero en caso de ser una corriente debe ser la corriente a través de una fuente de tensión.

La ganancia es el cociente entre la variable de salida y la variable de entrada:

$$\text{ganancia} = \frac{\Delta(\text{variable de salida})}{\Delta(\text{fuente de entrada})}$$

Dependiendo de la naturaleza de estas variables se obtienen ganancias propiamente dichas, que pueden ser de tensión (ambas variables tensiones) o de corriente (ambas variables corrientes), o bien transresistencias (salida tensión, entrada corriente) o transconductancias (salida corriente, entrada tensión).

La impedancia de entrada es la impedancia que "ve" la fuente independiente de entrada. La impedancia de salida es la que se obtendría aplicando un generador de prueba en el nodo de salida y midiendo la corriente absorbida desconectando el resto de fuentes independientes.

Los resultados se guardan bajo la sección "SMALL-SIGNAL CHARACTERISTICS" del fichero ".OUT". En el siguiente ejemplo se usa este análisis aplicado al circuito de la figura 2.15.

```
EX2_7A (Ejemplo de función de transferencia)

VA 1 0 1V

R1 1 0 2
R2 1 2 1
R3 2 0 1

.TF V(2) VA

.END
```




Fig. 2.15

Los resultados obtenidos en ".OUT" son:

```
**** SMALL-SIGNAL CHARACTERISTICS
```

```
V(2)/IEN = 5.000E-01
```

```
INPUT RESISTANCE AT IEN = 1.000E+00
```

```
OUTPUT RESISTANCE AT V(2) = 7.500E-01
```

Si en lugar de una fuente de tensión se tuviera el circuito de la figura 2.16 el *netlist* sería:

```
EX2_7B (Función de transferencia, transresistencia)
```

```
IEN 0 1 1A
```

```
R1 1 0 2
```

```
R2 1 2 1
```

```
R3 2 0 1
```

```
.TF V(2) IEN
```

```
.END
```




Fig. 2.16

El resultado obtenido es:

```

**** SMALL-SIGNAL CHARACTERISTICS

V(2)/IEN = 5.000E-01

INPUT RESISTANCE AT IEN = 1.000E+00

OUTPUT RESISTANCE AT V(2) = 7.500E-01

```

Obsérvese que en este caso la impedancia de salida es de $0,75\Omega$ en lugar de los $0,5\Omega$ del caso en que había una fuente de tensión, ya que una fuente de corriente "desactivada" es un circuito abierto mientras que una fuente de tensión es un cortocircuito.

Si se desea medir la ganancia de corriente entre dos ramas y la rama de salida no es una fuente de tensión se puede usar el siguiente "truco": añadir una fuente de tensión de valor cero voltios en la rama de salida. Ésta ha sido la técnica utilizada en el siguiente ejemplo, en el que se mide la ganancia de corriente entre la corriente a través del generador de entrada y la de la resistencia R3. Se ha añadido el generador VX al circuito con lo que queda tal como se ve en la figura 2.17. El fichero de entrada y el resultado son:

```

EX7_C (Función de transferencia, ganancia de corriente)

IEN 0 1 1A
R1 1 0 2
R2 1 2 1
R3 3 0 1
VX 2 3 0 ;generador añadido para medir la corriente

.TF I(VX) IEN

.END

```




Fig. 2.17

Resultado:

```

****  SMALL-SIGNAL CHARACTERISTICS

      I(VX)/IEN = 5.000E-01

      INPUT RESISTANCE AT IEN = 1.000E+00

      OUTPUT RESISTANCE AT I(VX) = 4.000E+00

```

Para llevar a cabo el análisis se utiliza el modelo linealizado en pequeña señal. Por ello habrá que tener en cuenta las limitaciones de este modelo, comentadas en el análisis AC.

El método que sigue el programa para realizar el análisis consiste en calcular el valor de las variables para el punto de reposo, a continuación modificar ligeramente el valor de la fuente de entrada y calcular el resto de variables. A partir de estos datos calcula los resultados. Esto es equivalente a realizar el análisis a frecuencia cero.

2.5 SENS (Análisis de sensibilidad)

Este análisis está encaminado a obtener la sensibilidad de las variables del circuito con respecto a diversos parámetros. La sintaxis es:

```
.SENS <variables de salida>*
```

Las variables de salida pueden ser una o varias. Su sintaxis es la misma que la utilizada en la orden ".PRINT" tal como se recoge en el apéndice F (variables de salida). Si se desea la sensibilidad de una

corriente ésta debe ser la corriente a través de un generador independiente de tensión por lo que, si es preciso, se debe insertar un generador de valor cero en la rama que interese.

Los resultados que proporciona este análisis son:

- sensibilidad:

$$\left(\frac{\Delta < \text{variable de salida} >}{\Delta < \text{parámetro} >} \right) \text{ (VOLTS/UNITY)}$$

- sensibilidad normalizada:

$$\left(\frac{\frac{\Delta < \text{variable de salida} >}{\Delta < \text{parámetro} >} \times 100}{< \text{parámetro} >} \right) \text{ (VOLTS/PERCENT)}$$

Si se tratara de una corriente, en lugar de *VOLTS* figuraría *AMPS*. Estos datos se guardan en el fichero de salida ".OUT".

Para realizar este análisis se usa el modelo linealizado por lo que, si existen elementos no lineales, los resultados no se pueden extrapolar para grandes variaciones del punto de reposo. La sensibilidad se calcula con respecto a los parámetros de los siguientes elementos:

- resistencias
- fuentes independientes de tensión y corriente
- interruptores controlados de tensión y corriente
- diodos
- transistores bipolares.

En el siguiente ejemplo se analiza la sensibilidad de la salida de un puente de impedancias frente a las impedancias que lo forman y la alimentación. El circuito es el mostrado en la figura 2.18 y el fichero fuente correspondiente es:

```

ANALISIS DE SENSIBILIDAD

VAL 1 0 5

R1 1 2 10K
R2 1 3 10K
R3 2 0 5K
R4 3 0 5K
.SENS V(3,2)

.END

```




Fig. 2.18

En el fichero de salida aparecerá una nueva sección, "DC SENSITIVITY ANALYSIS", con el siguiente contenido:

```

**** DC SENSITIVITY ANALYSIS    TEMPERATURE = 27.000 DEG C

*****

DC SENSITIVITIES OF OUTPUT V(3,2)

ELEMENT  ELEMENT  ELEMENT  NORMALIZED
NAME     VALUE    SENSITIVITY  SENSITIVITY
              (VOLTS/UNIT) (VOLTS/PERCENT)

R1       1.000E+04  1.111E-04    1.111E-02
R2       1.000E+04 -1.111E-04   -1.111E-02
R3       5.000E+03 -2.222E-04   -1.111E-02
R4       5.000E+03  2.222E-04    1.111E-02
VAL      5.000E+00  4.174E-18    2.087E-19

JOB CONCLUDED

```

Esto significa que si la resistencia R1 varía su valor en $1\text{k}\Omega$ la tensión de salida $V(3)-V(2)$ variará en:

$$\Delta V = 1,111 \cdot 10^{-4} \times 1000 = 0,1111 \text{ V} \quad (\text{compruébese})$$

o bien, si la resistencia R4 tienen una tolerancia del 5% (lo que implica un incremento de 250Ω) la máxima variación de la salida debido a la variación de R4 será:

$$\Delta V = 1,111 \cdot 10^{-2} \times 5 = 0,0555 \text{ V} \quad (\text{compruébese})$$

Se observa también que la sensibilidad de las ramas opuestas del puente tienen el mismo signo, por lo que si este puente se usa para medir variaciones de resistencia provocadas por determinadas magnitudes físicas (temperatura, desplazamientos, presiones, etc.) y se utiliza más de un elemento sensor, éstos deberán ocupar ramas opuestas.




Fig. 2.19

El análisis de sensibilidad tiene especial interés en circuitos donde existen diodos o transistores bipolares debido a la gran dispersión de valores que presentan estos dispositivos. En el siguiente análisis se estudia la sensibilidad de la tensión colector-emisor y la corriente de colector del circuito de la figura 2.19. El fichero fuente es:

Sensibilidad punto de reposo sin resistencia de emisor

```
.model BC108B NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=59.59 Bf=381.7 Ise=59.74f
+ Ne=1.522 Ikf=3.289 Xtb=1.5 Br=2.359 Isc=192.9p Nc=1.954
+ Ikr=7.807 Rc=1.427 Cjc=5.38p Mjc=.329 Vjc=.6218 Fc=.5 Cje=11.5p
+ Mje=.2718 Vje=.5 Tr=10n Tf=438p Itf=5.716 Xtf=14.51 Vtf=10)
```

VCC 4 0 6V ; alimentación

R1 4 2 390K

R2 2 0 195K

RC 4 30 1K


```
VTEST 30 3 0
Q1 3 2 0 BC108B

.SENS VCE(Q1) I(VTEST)
.END
```

El generador VTEST tiene la función de medir la corriente de colector. Como resultado en el fichero ".OUT" se obtienen, entre otros, los siguientes datos:

```
NODE VOLTAGE  NODE VOLTAGE  NODE VOLTAGE  NODE VOLTAGE
( 2)  .6914      ( 3)  3.0174      ( 4)  6.0000      ( 30) 3.0174
```

VOLTAGE SOURCE CURRENTS

NAME CURRENT

```
VCC      -2.996E-03
VTEST    2.983E-03
```

•
•
•

**** DC SENSITIVITY ANALYSIS TEMPERATURE = 27.000 DEG C

DC SENSITIVITIES OF OUTPUT VCE(Q1)

ELEMENT NAME	ELEMENT VALUE (VOLTS/UNIT)	ELEMENT SENSITIVITY (VOLTS/PERCENT)	NORMALIZED SENSITIVITY
R1	3.900E+05	1.056E-05	4.117E-02
R2	1.950E+05	-5.500E-06	-1.073E-02
RC	1.000E+03	-2.846E-03	-2.846E-02
VCC	6.000E+00	1.784E-01	1.071E-02
VTEST	0.000E+00	-9.540E-01	0.000E+00
Q1			
RB	0.000E+00	0.000E+00	0.000E+00
RC	1.427E+00	1.371E-04	1.956E-06
RE	0.000E+00	0.000E+00	0.000E+00
BF	3.817E+02	-5.966E-03	-2.277E-02
ISE	5.974E-14	1.285E+13	7.678E-03
BR	2.359E+00	3.844E-10	9.067E-12

ISC	1.929E-10	-3.034E+05	-5.853E-07
IS	7.049E-15	-8.027E+13	-5.659E-03
NE	1.522E+00	-8.861E+00	-1.349E-01
NC	1.954E+00	2.996E-05	5.853E-07
IKF	3.289E+00	-7.538E-04	-2.479E-05
IKR	7.807E+00	3.282E-16	2.562E-17
VAF	5.959E+01	1.791E-03	1.067E-03
VAR	0.000E+00	0.000E+00	0.000E+00
DC SENSITIVITIES OF OUTPUT I(VTEST)			
ELEMENT NAME	ELEMENT VALUE (AMPS/UNIT)	ELEMENT SENSITIVITY (AMPS/PERCENT)	NORMALIZED SENSITIVITY
R1	3.900E+05	-1.056E-08	-4.117E-05
R2	1.950E+05	5.500E-09	1.073E-05
RC	1.000E+03	-1.371E-07	-1.371E-06
VCC	6.000E+00	8.216E-04	4.929E-05
VTEST	0.000E+00	-4.596E-05	0.000E+00
Q1			
RB	0.000E+00	0.000E+00	0.000E+00
RC	1.427E+00	-1.371E-07	-1.956E-09
RE	0.000E+00	0.000E+00	0.000E+00
BF	3.817E+02	5.966E-06	2.277E-05
ISE	5.974E-14	-1.285E+10	-7.678E-06
BR	2.359E+00	-3.844E-13	-9.067E-15
ISC	1.929E-10	3.034E+02	5.853E-10
IS	7.049E-15	8.027E+10	5.659E-06
NE	1.522E+00	8.861E-03	1.349E-04
NC	1.954E+00	-2.996E-08	-5.853E-10
IKF	3.289E+00	7.538E-07	2.479E-08
IKR	7.807E+00	-3.282E-19	-2.562E-20
VAF	5.959E+01	-1.791E-06	-1.067E-06
VAR	0.000E+00	0.000E+00	0.000E+00
JOB CONCLUDED			

Se observa que la V_{CE} en reposo es de unos 3V y la corriente de colector de unos 3mA. La sensibilidad de la tensión V_{CE} con respecto a la variación porcentual de β es de $-2.2 \cdot 10^{-2}$ (la β está estrechamente relacionada con BF tal como se verá en el capítulo 4), no muy diferente a la sensibilidad con respecto a la resistencia RC. Pero los parámetros de los elementos semiconductores presentan varia-

ciones mucho más importantes que las de los elementos pasivos. Así, no es extraño que la β presente variaciones del 200%. Si esto sucediese la tensión V_{CE} podría sufrir una desviación de:

$$\Delta V_{CE} = -2,277 \cdot 10^{-2} \times 200 = -4,55V.$$

Esta variación llevaría al transistor a saturación. Para prevenir este efecto se suele añadir una resistencia en el emisor tal como se muestra en la figura 2.20. Si se simula este nuevo circuito utilizando el siguiente fichero de circuito:

Sensibilidad punto de reposo con resistencia de emisor

```
.model BC108B NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=59.59 Bf=381.7 Ise=59.74f
+      Ne=1.522 Ikf=3.289 Xtb=1.5 Br=2.359 Isc=192.9p Nc=1.954
+      Ikr=7.807 Rc=1.427 Cjc=5.38p Mjc=.329 Vjc=.6218 Fc=.5 Cje=11.5p
+      Mje=.2718 Vje=.5 Tr=10n Tf=438p Itf=5.716 Xtf=14.51 Vtf=10)
```

```
VCC 4 0 6V ; alimentación
```

```
R1 4 2 5.8K
```

```
R2 2 0 10K
```

```
RC 4 30 0.5K
```

```
RE 5 0 1K
```

```
VTEST 30 3 0
```

```
Q1 3 2 5 BC108B
```

```
.SENS VCE(Q1) I(VTEST)
```

```
.END
```




Fig. 2.20

se obtienen los siguientes resultados:

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(2)	3.7588	(3)	4.4723	(4)	6.0000	(5)	3.0660
(30)	4.4723						
VOLTAGE SOURCE CURRENTS							
NAME	CURRENT						
VCC	-3.442E-03						
VTEST	3.055E-03						
**** DC SENSITIVITY ANALYSIS TEMPERATURE = 27.000 DEG C							
DC SENSITIVITIES OF OUTPUT VCE(Q1)							
ELEMENT	ELEMENT	ELEMENT	NORMALIZED				
NAME	VALUE	SENSITIVITY	SENSITIVITY				
		(VOLTS/UNIT)	(VOLTS/PERCENT)				
R1	5.800E+03	3.587E-04	2.081E-02				
R2	1.000E+04	-2.024E-04	-2.024E-02				
RC	5.000E+02	-3.051E-03	-1.525E-02				
RE	1.000E+03	1.436E-03	1.436E-02				
VCC	6.000E+00	7.013E-02	4.208E-03				
VTEST	0.000E+00	-9.984E-01	0.000E+00				
Q1							
RB	0.000E+00	0.000E+00	0.000E+00				
RC	1.427E+00	4.788E-06	6.833E-08				
RE	0.000E+00	0.000E+00	0.000E+00				
BF	3.817E+02	-1.214E-04	-4.634E-04				
ISE	5.974E-14	2.571E+11	1.536E-04				
BR	2.359E+00	7.452E-12	1.758E-13				
ISC	1.929E-10	-5.883E+03	-1.135E-08				
IS	7.049E-15	-6.819E+12	-4.807E-04				
NE	1.522E+00	-1.776E-01	-2.703E-03				
NC	1.954E+00	5.808E-07	1.135E-08				

IKF	3.289E+00	-2.633E-05	-8.659E-07
IKR	7.807E+00	1.090E-17	8.508E-19
VAF	5.959E+01	0.000E+00	0.000E+00
DC SENSITIVITIES OF OUTPUT I(VTEST)			
ELEMENT NAME	ELEMENT VALUE	ELEMENT SENSITIVITY (AMPS/UNIT)	NORMALIZED SENSITIVITY (AMPS/PERCENT)
R1	5.800E+03	-2.386E-07	-1.384E-05
R2	1.000E+04	1.346E-07	1.346E-05
RC	5.000E+02	-3.510E-09	-1.755E-08
RE	1.000E+03	-2.995E-06	-2.995E-05
VCC	6.000E+00	6.186E-04	3.712E-05
VTEST	0.000E+00	-1.149E-06	0.000E+00
Q1			
RB	0.000E+00	0.000E+00	0.000E+00
RC	1.427E+00	-3.510E-09	-5.009E-11
RE	0.000E+00	0.000E+00	0.000E+00
BF	3.817E+02	9.455E-08	3.609E-07
ISE	5.974E-14	-2.002E+08	-1.196E-07
BR	2.359E+00	-5.802E-15	-1.369E-16
ISC	1.929E-10	4.581E+00	8.836E-12
IS	7.049E-15	4.699E+09	3.312E-07
NE	1.522E+00	1.383E-04	2.105E-06
NC	1.954E+00	-4.522E-10	-8.836E-12
IKF	3.289E+00	1.930E-08	6.348E-10
IKR	7.807E+00	-7.990E-21	-6.238E-22
VAF	5.959E+01	-1.367E-08	-8.147E-09
VAR	0.000E+00	0.000E+00	0.000E+00
JOB CONCLUDED			

En este caso la corriente de colector es de unos 3mA, como en el caso anterior, y la tensión V_{CE} es de 1,41V, o sea, que el transistor está mucho más cerca de saturación que antes. Pero ahora la sensibilidad con respecto a β es de $-4,63 \cdot 10^{-4}$ por lo que una variación de β del 200% producirá una variación de V_{CE} de:

$$\Delta V_{CE} = -4,63 \cdot 10^{-4} \times 200 = -0,093V.$$

Así pues la tensión V_{CE} disminuirá hasta 1,31V debido a una variación de β del 200%, pero esto no hará que el transistor entre en saturación.

En realidad estos cálculos no son del todo correctos porque, tal como se ha dicho, se usan los modelos linealizados y, en consecuencia, estos análisis no son válidos para grandes desplazamientos del punto de reposo. De todas formas se deduce que el segundo circuito es mucho más insensible a los parámetros del transistor que el primero.

2.6 FOUR (Análisis de Fourier)

Con este análisis se obtienen las componentes de Fourier de la variable o variables especificadas. Un análisis de este tipo debe ir precedido de un análisis transitorio ya que lo que en realidad realiza el programa es una Transformada Discreta de Fourier (DFT) y para ello debe muestrear la señal en el tiempo, lo que se corresponde con los resultados de un análisis transitorio.

La sintaxis de este análisis es la siguiente:

.FOUR <frecuencia de la fundamental> [número de armónicos] <variable>*

Para el cálculo se toman los valores de la variable obtenidos al final del análisis transitorio. Los valores que se usan para el cálculo son los correspondientes a los últimos "1/<frecuencia de la fundamental>" segundos. Esto significa que la señal que se considera es una señal como la que existe al final del análisis transitorio y periódica con frecuencia igual a la fundamental. Es claro que el valor de la frecuencia de la fundamental es clave para obtener resultados coherentes.

Los datos que se obtienen son: a) la componente continua; b) la componente de la fundamental (1^{er} armónico); c) la de los armónicos del 2^o al 9^o y d) la distorsión armónica total. Si se desean más componentes se han de obtener por medio del parámetro optativo [número de armónicos].

Los resultados se guardan en el fichero de salida ".OUT".

Por ejemplo, supóngase que se desea conocer el desarrollo en serie de Fourier de una señal cuadrada de 10V de amplitud y frecuencia 1kHz y se tiene pereza para calcularlo analíticamente. Se puede obtener una buena aproximación numérica realizando el análisis de Fourier del circuito mostrado en la figura 2.21.




Fig. 2.21

El fichero fuente será:

```

Análisis de Fourier

*generador
VI 1 0 10 PULSE( 0 10 0 1E-6 1E-6 0.5E-3 1E-3)
R1 1 0 10K

.TRAN 1U 1.5E-3
.FOUR 1E3 V(1)
.PROBE

.END

```

y en el fichero ".OUT" se obtendrá la siguiente sección con los resultados:

```

****   FOURIER ANALYSIS           TEMPERATURE = 27.000 DEG C

FOURIER COMPONENTS OF TRANSIENT RESPONSE V(1)

DC COMPONENT = 5.009999E+00

HARMONIC FREQUENCY  FOURIER  NORMALIZED PHASE NORMALIZED
NO              (HZ)  COMPONENT COMPONENT (DEG) PHASE (DEG)

  1      1.000E+03      6.366E+00   1.000E+00   1.796E+02   0.000E+00
  2      2.000E+03      2.000E-02   3.141E-03   8.928E+01  -9.036E+01
  3      3.000E+03      2.122E+00   3.333E-01   1.789E+02  -7.200E-01
  4      4.000E+03      2.000E-02   3.141E-03   8.856E+01  -9.108E+01
  5      5.000E+03      1.273E+00   2.000E-01   1.782E+02  -1.440E+00
  6      6.000E+03      2.000E-02   3.141E-03   8.784E+01  -9.180E+01
  7      7.000E+03      9.093E-01   1.428E-01   1.775E+02  -2.160E+00
  8      8.000E+03      2.000E-02   3.141E-03   8.712E+01  -9.252E+01
  9      9.000E+03      7.072E-01   1.111E-01   1.768E+02  -2.880E+00

TOTAL HARMONIC DISTORTION = 4.288115E+01 PERCENT

JOB CONCLUDED

```

Otra posibilidad para obtener estos resultados es utilizar el programa PROBE. Mediante este programa (del que se trata en el capítulo 5) se puede calcular la FFT cuando se está visualizando el resultado de un análisis transitorio. A este análisis se accede a través del menú *eje X* tal como se describe en el

apartado 5.5. Una ventaja que ofrece este segundo método es que la fracción de señal que se considera se determina con la opción *Restric_data*, por lo que la decisión de qué parte de señal es significativa se puede realizar después del análisis transitorio y una vez que ya se conoce la respuesta y se dispone de una representación gráfica de la misma. Por otra parte la información que se obtiene mediante PROBE es gráfica mientras que con ".FOUR" se obtiene una salida tabulada. También se ha de considerar que el programa PROBE es exclusivo de PSpice y no se encuentra en otras versiones de SPICE (aunque se suele contar con otras herramientas similares).

2.7 Otros análisis

En SPICE existen otros análisis además de los descritos hasta aquí. Pero se dará tan sólo una descripción de ellos debido al carácter introductorio de estos libro. Para obtener mayor información se puede acudir a la bibliografía especializada [GARIB95]. Los análisis pendientes son:

- Análisis estandar:
.NOISE: análisis de ruido. Precisa de un análisis ".AC" previo. Muchos dispositivos poseen un modelo de ruido que es el que se utilizará en este análisis.
- Análisis multi-ejecución (*multi-run*):
.TEMP:
Este análisis realizará los análisis normales descritos anteriormente pero a la temperatura o temperaturas indicadas mediante la siguiente sentencia:
.TEMP <lista de temperaturas>*
Cuando no se especifica un análisis de este tipo se considera que los valores de cada componente son los correspondientes a la temperatura TNOM, que es una variable que se fija mediante ".OPTIONS" tal como se describe en el capítulo 1. Por defecto TNOM vale 27° C. Al variar la temperatura los valores de los componentes varían según unas leyes incluidas en los modelos de cada componente y cuyas constantes se fijan por medio de parámetros de los modelos (consúltese el capítulo 3), o bien en la propia línea de declaración del componente si el componente es una resistencia, .

.STEP: similar al anterior, pero en lugar de variar la temperatura se puede variar otro parámetro. Es similar por otro lado al análisis ".DC".
- Análisis estadísticos:

.MC: análisis de Montecarlo. Proporciona información estadística. Se puede especificar la función de variación de los parámetros de los dispositivos

.WC: análisis de caso más desfavorable.

Estos dos últimos análisis hacen uso de la información sobre tolerancias proporcionada junto con los valores de los dispositivos tal como se verá en siguiente el capítulo.